Tutorial 31: Complément sur les listes

Identique au tutorial 30… mais pour les listes

Introduction

Pareil qu'auparavant mais appliqué aux listes, c'est parti

Nouvelles commandes

_adrlele

- calcule l'adresse de l'élément de la liste

_createrlist
- crée une liste

_getltoop1

- élément dans Op1

Le code

Les deux programmes de Ian:

.NOLIST

#define equ .equ

#define EQU .equ

#define end .end

#include "ti83asm.inc"

#include "tokens.inc"

.LIST

_createrlist .equ 445Ah

rlstobj .equ 01h

.org 9327h

     call _zerooop1         ; clear op1

     ld hl,listname         ; load var name for list IAN into op1

     ld de,op1              ;

     ld bc,6                ;

     ldir                   ;

     call _chkfindsym       ; look up list

     call nc,_delvar        ; if it is there delete it

     ld hl,2                ; number of elements (1-999)

     call _createrlist      ; create list

     push de                ; save list address

     ld hl,1                ; element 1

     call _adrlele          ; calculate element address

     ex de,hl               ; de -> element address

     ld hl,element1         ; hl -> num

     ld bc,9                ; 9 bytes

     ldir                   ; copy num to element

     pop de                 ; restore list address

     ld hl,2                ; element 2

     call _adrlele          ; calculate element address

     ex de,hl               ; de -> element address

     ld hl,element2         ; hl -> num

     ld bc,9                ; 9 bytes

     ldir                   ; copy num to element

     ret

listname:

     .db rlstobj,tvarlst,"IAN",$00  ; the user list IAN

element1:

     .db $00,$80,$75,$00,$00,$00,$00,$00,$00  ; 7.5

element2:

     .db $80,$82,$73,$45,$67,$23,$89,$00,$00  ; -734.5672389

.end

END

C'est familier non? Cela doit! C'est presque identique au programme des matrices. Pour la suite… c'est aussi identique

.NOLIST

#define equ .equ

#define EQU .equ

#define end .end

#include "ti83asm.inc"

#include "tokens.inc"

.LIST

_errundefined .equ 467Bh

rlstobj .equ 01h

.org 9327h

     call _zerooop1      ; clear op1

     ld hl,listname      ; copy var name for list IAN into op1

     ld de,op1           ;

     ld bc,6             ;

     ldir                ;

     call _chkfindsym    ; look up list

     jp c,_errundefined  ; error if not found

     push de             ;

     pop hl              ; hl -> list address

     call _ldhlind       ; hl -> list length

     push hl             ;

     pop bc              ; bc -> list length

     ld hl,1             ; hl -> element number

lloop:

     push bc             ; store count

     push hl             ; store element num

     push de             ; store list address

     call _getltoop1     ; get element to op1

     ld a,16             ; format number for display

     call _formreal      ;

     ld hl,16            ;

     sbc hl,bc           ;

     ld a,l              ;

     ld (curcol),a       ;

     ld hl,op3           ;

     call _puts          ; display it

     pop de              ; restore list address

     pop hl              ; restore element number

     inc hl              ; increment element number

     pop bc              ; restore count

     dec bc              ; decrement count

     ld a,b              ; see if count is at 0

     or c                ;

     jr nz,lloop         ;

     ret

listname:

     .db rlstobj,tvarlst,"IAN",$00  ; the user list IAN

.end

END

Les listes sont totalement étudiées!

Conclusion

Félicitations!! Voici la fin de la deuxième série de tutorials !! J'apprécierai  vraiment un e-mail sur vos impressions de ces tutorials. J'ai besoin de commentaires pour améliorer ces pages. Ce que j'aimerais faire dans le futur: liaison, syntaxe, graphisme, exécution de programmes basic, fonctions de math en ASM…

Toutes les informations pour créer des jeux sont là. Je travaille aussi sur d'autres commandes, comme les erreurs. Merci de lire mes tutorials… à plus

James

Merci à Ian Graf

