Tutorial 53: la routine de Movax expliquée

Un code à vous casser la tête!

Introduction

Bien, retour au tutorial 15, à propos de la souris sur TI - une  petite souris qu'utilise la routine de Movax. Nous ne savions pas grand chose et bien maintenant… nous savons! Regardons cette routine en détail.

Code

; SPRXOR 

; Xor 8x8 sprite a=x, e=y, bc=sprite address

SPRXOR:

        push    bc              ; Save sprite address

;====   Calculate the address in graphbuf   ====

        ld      hl,0            ; Do y*12

        ld      d,0

        add     hl,de

        add     hl,de

        add     hl,de

Ok hl=0 et de= coordonnée Y. Movax les additionne 3 fois. Donc nous avons з fois le coordonnée Y.

add     hl,hl

add     hl,hl

Maintenant il les additionne deux fois, nous avons donc 12 fois Y. Vous devez savoir ça de la routine d'Olle sur le changement de pixel.

        ld      d,0             ; Do x/8

        ld      e,a

        srl     e

        srl     e

        srl     e

        add     hl,de

        ld      de,8e29h

        add     hl,de           ; Add address to graphbuf

        ld      b,00000111b     ; Get the remainder of x/8

        and     b
Même principe que pour le code d'Olle.

        cp      0               ; Is this sprite aligned to 8*n,y?

        jp      z,ALIGN

Une sprite aligné est une sprite dont le côté gauche est aligné avec le premier bit de chaque octet. Exemple (8,15) (16,34) (24,1) etc

;====   Non aligned sprite blit starts here   ====

        pop     ix              ; ix->sprite
Souvenez-vous que l'adresse de la routine est sauvegardée.

ld      d,a             ; d=how many bits to shift each line

a contient le reste de x/18

ld      e,8             ; Line loop

LILOP:  ld      b,(ix+0)        ; Get sprite data

        ld      c,0             ; Shift loop

        push    de

SHLOP:  srl     b

        rr      c

        dec     d

        jp      nz,SHLOP

        pop     de

Oh cette partie m'a été très difficile à comprendre - J'ai demandé au programmeur, voici ce qu'il m'a dit:

Nous commençons avec la donnée de la sprite que nous voulons décaler en b. Après le slr (décalage logique à droite) nous voulons attraper le bit qui a été décalé dehors et le placer dans c. On le fait avec rr (rotation à droite). Après on décrémente d tant que le compteur est dans la boucle et que d est différent de 0 on retourne à SHLOP

Maintenant c'est un peu plus facile à comprendre Juste une note, slr met le bit décalé dans la retenue et rr prend la retenue et fait une rotation. Peut-être avons-nous besoin d'un exemple…

Elargissement

Disons que la première ligne de la sprite est {0,0,1,1,0,1,0,1} et que vous voulez la placer à (31,5), le reste de 31/8 est 7. Le programme doit donc décalé la sprite de 7 bits. En utilisant d comme compteur de décrémentation, slr décale le tout d'un cran vers la droite et met le bit décalé dans la retenue. Le rr prend la retenue et l'insère le plus à gauche de ses bits. Donc au début vous avez:

B = 00110101

C = 00000000

Après le code est répété et vous avez:

B = 00000000

C = 01101010
La ligne de la sprite est correctement décalée! Retour au code…

        ld      a,b             ; Write line to graphbuf

        xor     (hl)

        ld      (hl),a

        inc     hl

        ld      a,c

        xor     (hl)

        ld      (hl),a
On met la ligne décalée de la sprite dans plotsscreen

        ld      bc,11           ; Calculate next line address

        add     hl,bc

        inc     ix              ; Inc spritepointer
On saute seulement de 11 octets comme nous l'avons déjà fait pour les deux autres sprites. Donc la nouvelle adresse de plotsscreen est calculée et IX est incrémenté au point de la ligne suivante de la sprite.

        dec     e

        jp      nz,LILOP        ; Next line

        jp      DONE1

;====   Aligned sprite blit starts here   ====

ALIGN:                          ; Blit an aligned sprite to graphbuf

        pop     de              ; de->sprite

        ld      b,8

ALOP1:  ld      a,(de)

        xor     (hl)

        ld      (hl),a

        inc     de

        push    bc

        ld      bc,12

        add     hl,bc

        pop     bc

        djnz    ALOP1
Vous devez être capable de comprendre ceci sans explications :)…

DONE1:

        ret

; SPRXOR 

; Z80 Sprite83  movax 

Quel beau code!

Conclusion

Ok c'est fait, je l'ai FINI! 53 tutorials, la version з.0 arrive à sa fin! Félicitations si vous avez suivi tous les tutorials les uns après les autres, vous pouvez être fier de vous! La prochaine version, j'espère avoir des trucs sur la transmission par câble, plus de fonctions graphiques, quelques fonctions non documentées. j'aimerais jeter un œil sur des algorithmes, des jeux de physiques par exemple, comment implémenter la gravité, la friction etc. Donnez moi des idées!

